
Объектно-ориентированное программирование

Лабораторная работа № 9
Создание компонентов Delphi
1. Теоретическая часть

1.1. Введение в создание компонентов Delphi 
При разработке приложений с помощью Borland Delphi создавать новые компоненты удобно по следующим причинам: 

1. Простота использования. Компонент помещается на форму, и для него необходимо устанавливать значения свойств и писать код обработчиков событий. Поэтому если в проекте какое-либо сочетание элементов управления и обработчиков связанных с ними событий встречается в двух и более местах, то имеет смысл подумать о создании соответствующего компонента, что гарантированно сэкономит усилия при разработке приложения. 

2. Простая организация групповой разработки проекта. При групповой разработке отдельные части проекта можно определить как компоненты и поручить эту работу разным программистам. Компоненты можно отладить отдельно от приложения, что сделать достаточно легко. 

3. Простой и эффективный способ обмена кодом с другими программистами. Имеется немало сайтов, например http://www.torry.net/, где можно найти свободно распространяемые компоненты или приобрести их за символическую плату. 

1.2. Пакеты компонентов 
В Delphi компоненты хранятся в специальных файлах с расширением *.bpl, которые называются пакетами (packages). Список используемых пакетов компонентов можно просмотреть с помощью команды меню Component/Install Packages (рис.1). 

Рис. 1

Пакеты содержат откомпилированный код компонентов и представляют собой DLL (динамически загружаемые библиотеки) особого типа. Каждый пакет может содержать множество компонентов. В то же время, каждый компонент может находиться только в одном пакете. Пакеты создаются в результате компиляции исходного кода компонентов.
При помощи этого диалогового окна можно добавить новый пакет (кнопка Add), удалить имеющийся (кнопка Remove). Удаление означает не физическое удаление файла с диска, а удаление ссылки из среды разработки на данный пакет. При добавлении нового пакета компоненты, хранящиеся в нем, появляются на палитре, а при удалении – наоборот, исчезают. Пакет можно не удалять, а «спрятать» его содержимое на этапе разработки посредством снятия отметки напротив имени пакета в списке. Можно также просмотреть компоненты и их пиктограммы (кнопка Components). И наконец, можно отредактировать добавленные пользователем пакеты (кнопка Edit). Пакеты, поставляемые вместе с Delphi, редактировать нельзя (кнопка Edit в этом случае недоступна). 

В диалоговом окне рис. 1 можно также задать способ использования пакетов. По этому признаку пакеты компонентов делятся на три типа:

1. Runtime package (пакет, работающий во время выполнения приложения).
2. Designtime package (пакет, используемый во время разработки). 
3. Designtime and Runtime (пакет, используемый на обоих этапах – и разработки, и выполнения).

Все они представляют собой DLL. 

Runtime-пакеты (файлы с расширением *.bpl) поставляются конечному пользователю вместе с проектом, если проект был скомпилирован с включенным флажком Build with runtime packages. Само приложение (*.exe или *.dll) в этом случае получается небольшим, потому что в нем будет храниться только код программы. Все компоненты, которые установлены на форме, не попадут в исполняемый файл. Когда программа будет запускаться, она будет подгружать эти компоненты из файлов пакетов, перечисленных в строке чуть ниже вышеупомянутого флажка. Так можно уменьшить размер приложения, но при его переносе на другой компьютер, вместе с программой надо передавать довольно объемные *.bpl-файлы. Их нужно скопировать в папку Windows\System32.

Разработка проекта с runtime-пакетами дает преимущество в объеме используемых файлов, если только проект включает несколько модулей Delphi (по оценкам специалистов не менее пяти). При совместной работе этих модулей достигается экономия ресурсов операционной системы, поскольку один загруженный в ОП пакет обслуживает несколько модулей. 

Design-time-пакеты (файлы с расширением *.dcp) используются только на этапе разработки приложения. Флажок Build with runtime packages в этом случае сброшен. Во время разработки такие пакеты поддерживают создание компонентов на форме. Программа становится полноценной и не нуждается в дополнительных *.bpl-файлах. В скомпилированный проект Delphi включает код не из пакета компонентов, а из отдельно создаваемых *.dcu-файлов. При этом *.dcp-файлы генерируются из *.dcu-файлов.

Пакеты третьего типа содержат компоненты, доступные как на этапе разработки, так и в режиме выполнения программы. Флажок Build with runtime packages в этом случае также сброшен. Данный способ использования пакетов устанавливается в Delphi по умолчанию и используется наиболее часто.
1.3. Этапы и способы создания нового компонента
Процесс построения нового компонента значительно отличается от процесса разработки нового приложения. В ходе разработки приложения активно используется уже существующая в системе Delphi библиотека визуальных компонентов VCL (Visual Component Library). При создании пользовательского компонента эта библиотека активно изменяется и пополняется. 

Новый компонент обычно разрабатывается не с нуля, а на основе уже существующего класса, функции которого детализируются или расширяются путем изменения имеющихся и добавления новых свойств и методов. Разработка компонента, в отличие от визуального конструирования обычного приложения, сводится в основном к написанию исходного кода модуля компонента, для чего совершенно необходимо глубокое знание принципов объектно-ориентированного программирования и владение ими на уровне используемой системы программирования.
Цикл разработки нового компонента включает следующие этапы:
1. Анализ требований к новому компоненту. На данном этапе, исходя из исходных условий, которые формулируются как задание на разработку, необходимо тщательно продумать, какие функции должны реализовываться компонентом, какими характерными свойствами и методами он должен обладать.
2. Выбор подходящего родительского класса. По результатам предыдущего этапа нужно выбрать из VCL класс, функции которого наиболее близко подходят к требуемым функциям нового компонента. 

3. Программирование компонента. На данном этапе в модуле компонента реализуются свойства, методы и события компонента.

4. Установка компонента в палитру компонентов системы Delphi.
5. Тестирование компонента в отдельно создаваемом тестовом приложении.
6. Создание значка (ресурса) для нового компонента. 
7. Создание справки компонента.
По объему и трудоемкости, наиболее сложным из всех перечисленных является третий этап. Модуль компонента может быть создан двумя способами:
· ручным;
· с помощью Мастера компонентов.

Ручной способ создания модуля компонента заключается в подготовке нового модуля посредством команды меню File/New/Unit. Далее следует ввести программный код модуля, аналогичный сгенерированному Мастером. 
Наиболее часто применяется второй способ, поскольку использование Мастера позволяет упростить и ускорить создание модуля компонента. Мастер позволяет создать модуль, в котором будет объявлен минимально функциональный, т. е. родительский компонент, готовый к установке в среду разработки. После этого можно приступить непосредственно к реализации функций нового компонента, добавляя к родительскому новые свойства, методы и события.
Чтобы вызвать Мастер создания модуля компонента, можно воспользоваться одним из двух способов, выполнив в главном меню Delphi одну из следующих команд:

· команду Component/New Component;
· команду File/New/Other.
Во втором случае в открывшемся диалоговом окне New Items (рис. 2) следует выполнить двойной щелчок по значку Component. 

Рис. 2
В любом случае будет открыто диалоговое окно New Component мастера компонентов, представленное на рис. 3. Назначение и использование элементов этого окна поясняются в следующем подразделе.

Рис. 3
1.4. Создание простейшего компонента 
При написании нового компонента необходимо ясно представлять, что компонент создается для программистов, разрабатывающих приложения Windows, а не для конечных пользователей. При этом желательно, чтобы программист не вникал в детали реализации компонента, а просто пользовался экспонируемыми им свойствами и событиями. Это достигается очень тщательным тестированием. Новый компонент необходимо тестировать даже в ситуациях, для работы в которых он явно не предназначен. 

Рассмотрим процесс создания компонента на простом примере. 
Пример 1. Необходимо создать компонент, который будет формировать список принтеров, установленных в операционной системе.

Действия по созданию компонента разделим на шаги, упомянутые в п. 1.3.
1. Анализ требований к компоненту
Исходя из того, что проектируемый компонент должен отображать список принтеров необходимо, чтобы он удовлетворял следующим требованиям:
· при текущей работе не занимал на форме много места;

· отображал сведения о нескольких принтерах и давал возможность выбрать один их них;

· программист имел возможность задать размер и положение компонента;

· реагировал на основные события мыши и клавиатуры;

· на стадии разработки содержал «живые данные».

2. Выбор подходящего родительского класса
Анализ требований п.1 позволяет сделать вывод, что по выполняемым функциям для нового компонента в качестве родительского наиболее подходящим является стандартный компонент ComboBox (класс TComboBox). Понадобятся все его свойства, методы и события, а также те, которые следует добавить на стадии программирования.
3. Программирование компонента
Для реализации данного этапа будет использоваться Мастер компонентов. Выполните в главном меню Delphi команду Component / NewComponent. В открывшемся диалоговом окне New Component (рис. 3) введите следующие значения:

· Ancestor Type (тип родительского класса): выберите из списка TComboBox;
Примечание. Мастер предлагает в качестве наследника два класса TComboBox. Один из модуля StdCtrls, второй из QStdCtrls. Первый наследник относится к иерархии классов VCL, второй к CLX
. В данном примере (и во всех последующих) будет рассматриваться создание компонентов для VCL.
· Class Name (имя класса, реализующего новый компонент): согласно правилам, имя каждого класса в Delphi должно начинаться буквой Т. При этом имя вновь создаваемого класса должно отражать содержание компонента и ни в коем случае не совпадать с именем уже зарегистрированного компонента. Выберем имя «TPrinterList».

· Palette Page (вкладка палитры): вкладка в палитре компонентов, на которую Delphi поместит созданный компонент. Можно выбрать любую из имеющихся вкладок либо ввести имя новой, которую Delphi создаст. Поместим компонент па новую вкладку с именем New.
Примечание. При выполнении работы с целью разделения пользователей рекомендуется в имени вкладки использовать одну последнюю (для групп с трехзначным номером) цифру номера группы или две последних цифры (для групп с четырехзначным номером), например New0, New1 и т. д.
· Unit File Name (имя файла модуля): здесь для имени файла можно оставить значение по умолчанию и выбрать текущую папку, в которой будет сохранен исходный код модуля компонента.

· Search Path (последовательность поиска): список папок, в которых Delphi будет искать исходные файлы компонентов.
После ввода всех значений окно New Component примет вид рис. 4

Рис. 4

Нажмите в этом окне ОК. Delphi создаст заготовку кода модуля нового компонента: 
unit PrinterList;

   interface

   uses

  SysUtils, Classes, Controls, StdCtrls;

    type

  TPrinterList = class(TComboBox)

  private

    { Private declarations }

  protected

    { Protected declarations }

  public

    { Public declarations }

  published

    { Published declarations }

  end;

procedure Register;

implementation

procedure Register;

begin

  RegisterComponents('New', [TPrinterList]);

end;

end.

В самом новом классе объявлены четыре стандартные секции, а также определена процедура Register, которая вызывается средой разработки Delphi при инсталляции данного модуля как компонента. Она содержит имя страницы на палитре, куда помещается данный компонент, и в квадратных скобках – имя класса.
Полученный компонент пока ничем — ни внешним видом, ни поведением — не отличает​ся от своего предка, компонента ComboBox. Компонент PrinterList должен отображать список всех установленных принтеров. Это значит, что сразу после создания объекта класса TPrinterList приложение должно получить список установленных в системе принтеров и поместить их имена в выпадающий список компонента на форме. Здесь необходимо определить, на каком участке программы компонента можно реализовать это действие. Возможны следующие варианты:

· в конструкторе Create. Однако этот способ не подходит, потому что в момент помещения компонента на форму у него еще нет родительского окна (не установлено значение свойства ParentWindow), вследствие чего свойство Items еще не доступно;

· в инициализационном методе Loaded. Этот вариант также нельзя использовать, потому что метод Loaded вызывается при загрузке компонента в память и при помещении компонента на форму его список окажется пуст;

· в методе CreateWnd, унаследованном от класса TCustomComboBox. Это решение наиболее целесообразно, потому что этот метод вызывается именно в момент создания элемента интерфейса оконного типа.
Таким образом, в модуле компонента нужно переопределить метод CreateWnd. Для этого следует указать в секции public заготовки модуля заголовок метода:

procedure CreateWnd; override;

(с обязательной директивой override!) и реализовать его в секции реализации:

procedure TPrinterList.CreateWnd;

   begin

     inherited;

      Items.Assign(Printer().Printers);

    end;  

В теле этого метода вызывается функция Assign() доступа к принтерам, реализованная в модуле Printers, поэтому следует добавить в секцию Uses модуля PrinterList имя модуля Printers. Дополненный текст модуля компонента должен иметь следующий вид:
unit PrinterList;

   interface

uses

     SysUtils, Classes, Controls, StdCtrls, Printers;

type

     TPrinterList = class(TComboBox)

  private

    { Private declarations }

  protected

    { Protected declarations }

  public

    { Public declarations }

     procedure CreateWnd; override;

  published

    { Published declarations }

  end;

      procedure Register;

implementation

  procedure TPrinterList.CreateWnd;

   begin

     inherited;

      Items.Assign(Printer().Printers);

    end;  

procedure Register;

begin

  RegisterComponents('New', [TPrinterList]);

end;

end.

Сохраните модуль компонента командой File/Save As в рабочем каталоге, используя в качестве имени модуля имя, указанное в строке Unit File Name диалогового окна New Component.
4. Установка компонента в палитру
Для выполнения данной операции выполните команду меню Component/Install Component. В появившемся диалоговом окне Install Component перейдите на вкладку Into New Package («Поместить в новый пакет», рис. 5). 

Рис. 5

В заполненных полях оставьте предложенные значения. В поле Unit file name с помощью кнопки Browse введите полное имя файла модуля компонента. В поле Package file name введите полное имя файла нового пакета, например, «MyPackage», а в поле Package description — понятное описание пакета, например, «Пример 1 пользовательского компонента». Нажмите ОК и подтвердите создание пакета нажатием кнопки Yes в появившемся окне сообщения.

Delphi скомпилирует исходный файл модуля, создаст пакет и установит в него новый компонент. В результате в палитре компонентов должна появиться вкладка New, содержащая единственный компонент PrintersList, имя которого отображается во всплывающей подсказке. Файл пакета находится в той же папке, в которую вы поместили файл модуля (Unit File Name в окне Мастера компонентов), имеет полное имя Package file name и расширение *.dpk. Cодержимое пакета можно просмотреть в окне, которое открывается по команде File/Open Project. В диалоговом окне открытия файла пакета следует указать полное имя из строки Package file name окна Install Component и тип файла *.dpk.
Выполните эту команду для пакета MyPackage (рис. 6).

Рис. 6
Файл пакета типа *.dpk при разработке компонентов играет ту же роль, что и файл проекта для обычных приложений. Система использует его при компиляции и сборке пакета в двоичный файл *.bpl. Сам проект состоит из двух частей:
· Contains – список модулей, входящих в пакет. В примере в пакете только один модуль и его файл PrinterList.pas.
· Requires – список библиотек, необходимых для компиляции данного пакета. Здесь указывается список файлов *.dcp
, необходимых для компиляции.

В верхней части окна находится панель со следующими кнопками:

· Compile – компиляция пакета;
· Add – добавление нового модуля в пакет;

· Remove – удаление модуля;

· Install – установка пакета в систему (при нажатии этой кнопки пакет, при необходимости, будет перекомпилирован);

· Options – настройка параметров компиляции пакета. 

Нажмите кнопку Options. На экране откроется диалоговое окно (рис. 7) настройки параметров компиляции пакета.

Рис. 7
Все вкладки этого окна, кроме вкладки Description, аналогичны вкладкам окна настройки компиляции обычного приложения. На вкладке Description интерес представляет раздел Usage options, в котором задается тип нового пакета (см. п. 1.2). 
5. Тестирование компонента
Командой File/Close all закройте все окна, связанные с компонентом и пакетом. Создайте новое приложение, поместите на форму компонент PrinterList с вкладки New, а также метку с текстом «Список установленных принтеров» и кнопку закрытия приложения. Соберите и запустите приложение. Проверьте, как новый компонент реагирует на события, поступающие от клавиатуры и мыши.
6. Создание значка компонента
Чтобы компонент было легко найти в палитре, у него должен быть свой собственный значок. Сейчас созданный компонент наследует все, включая значок, от своего предка — компонента ComboBox. Теперь поменяем значок компонента PrinterList на палитре. 
Для создания и изменения значков предназначен встроенный в систему Delphi Редактор изображений Image Editor. Откройте редактор Image Editor командой Tools/Image Editor. В окне редактора выполните команду File/New/Component Resource File (*.dcr). Программа создаст новое окно Untitled1.dcr, содержащее дерево, в котором имеется пока только один элемент Contents. Для создания нового элемента откройте контекстное меню в окне Untitled1.dcr и в нем выполните команду New/Bitmap. Откроется окно свойств Bitmap Properties создаваемой картинки (рис. 8).

Рис. 8

Параметры размера картинки (Width и Height), задаваемые по умолчанию (32 пиксела), следует изменить на 24х24 – такой размер обязаны иметь значки всех компонентов. Для параметра Colors (Цветовой режим) можно оставить значение VGA, потому что для значка 16 цветов достаточно. Далее нажмите кнопку ОК в окне Bitmap Properties. В результате в окне Untitled1.dcr в структуре дерева элементов ресурсов появится раздел Bitmap и в нем новая картинка Bitmap1. Откройте контекстное меню на элементе Bitmap1 и в нем выполните команду Rename. Переименуйте картинку, дав ей имя класса нового компонента TPINTERLIST (прописными буквами - рис. 9).
Примечание. Картинка обязательно должна иметь то же имя, что и класс, которому принадлежит компонент. Это связано с тем, что в исходном коде приложения может быть несколько компонентов. Delphi по имени картинки будет определять, к какому компоненту (и, следовательно, классу) она относится. 

Рис. 9

Далее выполните двойной щелчок по элементу TPINTERLIST. Откроется окно графического редактора, в котором будут доступны стандартные инструменты, похожие на инструменты редактора Paint. Подобрав удобный масштаб, нарисуйте какое-либо изображение, которое и будет значком компонента.

После этого закройте окно Untitled1.dcr. На запрос системы о сохранении файла ресурсов откройте папку с исходным кодом модуля компонента и задайте имя, совпадающее с именем модуля компонента, но с расширением *.dcr - PrinterList.dcr.
Всех выполненных действий, однако, недостаточно для замены значка компонента на новый значок. Если посмотреть на палитру компонентов, то можно увидеть, что на вкладке New по-прежнему сохраняется прежний значок компонента. Если перезапустить Delphi или даже перезагрузить операционную систему, старый значок по-прежнему останется на палитре. Для того чтобы поменять значок, требуется повторная регистрация компонента. Для этого необходимо выполнить следующие действия: 

1) командой меню File/Open… или File/Open Project… открыть окно с пакетом MyPackage.dpk, содержащим данный компонент (рис. 10).

Рис. 10

2) в окне пакета нажать кнопку Remove. В возникающем диалоговом окне Remove From Project отметить исходный модуль компонента, который необходимо удалить, в нашем случае – PrinterList.pas. После этого следует нажать кнопку OK и подтвердить удаление в появляющемся окне; 

3) перекомпилировать пакет компонентов, нажав кнопку Compile. Это действие заставит систему удалить из оболочки прежний компонент. Если все было выполнено правильно, то появится сообщение об обновлении палитры компонентов;
4) нажать кнопку Add и в диалоге инсталляции нового компонента с помощью кнопки Browse задать путь к модулю компонента PrinterList.pas. После нажатия кнопки OK в окне пакета появятся как модуль *.pas, так и ссылка на файл *.dcr (рис. 11);

Рис. 11

5) снова нажать кнопку Compile. После компиляции пакета появится сообщение, что компонент PrinterList был инсталлирован в палитре компонентов и теперь его значок соответствует нарисованному. 
Создание справки компонента рассматривается в п. 1.7 теоретической части.
1.5. Добавление методов и событий в компонент
Созданный в примере 1 компонент PrinterList отличается от своего прототипа только переопределением одного из методов родительского компонента. Такой способ изменения поведения стандартного компонента и реализации дополнительных функций нового компонента является очевидным и достаточно простым. Другой способ состоит в том, что поведение стандартного компонента улучшается путем добавления к нему собственных методов и (или) переопределения обработчиков стандартных событий.
Рассмотрим первый способ на следующем примере. 
Пример 2. Создать компонент, который будет конвертировать текущее содержимое поля ввода в заглавные буквы, и подсчитывать количество вхождений задаваемого символа в содержимом поля ввода.
Из условия примера следует, что для нового компонента в качестве родительского наиболее подходящим является стандартный компонент Edit класса TEdit. Пусть новый компонент называется NewEdit. Для реализации возлагаемых на него функций в модуль компонента добавим два метода:
· процедуру ToUpper, которая будет преобразовывать текущее содержимое поля ввода в заглавные буквы;

· функцию с именем GetThisCharCount, которая будет возвращать количество вхождений задаваемого символа в содержимом поля ввода.
Выполнив действия пункта 3 в примере 1, создайте заготовку модуля нового компонента и добавьте в ее интерфейсную часть описания новых методов ToUpper и GetThisCharCount:
unit NewEdit;

  interface

   uses

  SysUtils, Classes, Controls, StdCtrls;

type

  TNewEdit = class(TEdit)

  private

    { Private declarations }

  protected

    { Protected declarations }

  public

  procedure ToUpper;

  function GetThisCharCount(C: char): Integer;

    { Public declarations }

  published

    { Published declarations }

  end;

procedure Register;

implementation
…
В раздел реализации включите программные коды новых методов, которые могут иметь следующий вид
:
  procedure TNewEdit.ToUpper;

   begin

     Text := Uppercase(Text); 

   end;

  function TNewEdit.GetThisCharCount(C: char): Integer; 
   var

     i : Integer; 

     begin

       Result := 0;

           for i := 1 to Length(Text) do
            if Text[i] = C then Inc(Result)

     end;
Установите компонент NewEdit на вкладку New палитры компонентов Delphi, выполнив действия пункта 4 примера 1. 
Выполните тестирование нового компонента, для чего создайте новое приложение, поместите на форму компонент NewEdit, компонент Edit для ввода задаваемого символа и три кнопки. Поля ввода с помощью компонентов Label снабдите заголовками, например: «Исходная строка», «Заданный символ». Кнопкам также можно дать названия, например: «Перевод в заглавные», «Кол-во вхождений», «Выход». Для первых двух кнопок подготовьте обработчики события OnClick так, чтобы каждая из них вызывала один из добавленных методов:
procedure TForm1.Button1Click(Sender: TObject);

   begin

     NewEdit1.ToUpper;

   end;

procedure TForm1.Button2Click(Sender: TObject);

   begin
    ShowMessage('Количество вхождений заданного символа '+Edit1.Text[1]+' = ' +

    IntToStr(NewEdit1.GetThisCharCount(Edit1.Text[1]))) ;

    NewEdit1.Text:=''; Edit1.Text:=''

  end;
Запустите приложение и проверьте правильность выполнения функций нового компонента NewEdit. 
Выполнив действия пункта 6 (пример 1), подготовьте и зарегистрируйте в среде Delphi значок для компонента NewEdit. 
При создании пользовательского компонента чаще всего от него требуется такая же реакция на события, как и от того компонента, на основе которого он разработан. Если родительским классом компонента выбран класс, реализующий, например, один из стандартных компонентов (класс Edit в примере 2), то специально заботиться о событиях не нужно: все события, доступные компоненту Edit, будут доступны и компоненту NewEdit.
Задача усложняется в следующих случаях: 
1. Если класс, реализующий новый компонент, является наследником некоторого абстрактного класса (например, можно было бы произвести класс TNewEdit от класса TCustomEdit – предка TEdit). 
2. Если требуется, чтобы новый компонент реагировал на некоторое событие не так, как его предок и эта реакция реализовывалась бы самим компонентом, а не средствами приложения, в котором он будет использоваться.
В первом случае в распоряжении нового класса по умолчанию никаких событий не будет. Если требуется, чтобы новый компонент получил доступ к событиям, например, OnClick и OnKeyDown, то необходимо объявить их как свойства в разделе published:
type

  TNewEdit = class(TCustomEdit)

  private

    { Private declarations }

  public

    { Public declarations }
  published
    property OnClick;

    property OnKeyDown;
  end;
После этого события OnClick и OnKeyDown появятся на вкладке Events Инспектора объектов, и можно будет обычным образом программировать их обработку в модуле компонента. 

Во втором случае необходимо просто переопределить в модуле компонента метод-обработчик требуемого события. Рассмотрим данный случай на следующем примере. 
Пример 3. Создать компонент, который будет представляться кнопкой, подсчитывающей количество нажатий на себя и отображающей его в своей надписи. В остальном новая кнопка должна реагировать на нажатие так же, как обычная. 
Поскольку новый компонент должен представляться кнопкой, то целесообразно в качестве базового класса выбрать класс TButton. В создаваемом компоненте будут использоваться его конструктор и обработчик события OnClick. Новому компоненту присвоим имя New1Button. 

Выполните все действия по созданию заготовки модуля нового компонента и добавьте описание нового класса TNew1Button следующими строками:
· в разделе private введите поле FCount:Integer для хранения количества нажатий на кнопку;
· в разделе public укажите заголовки переопределяемых методов - конструктора базового класса Create и обработчика события OnClick:
constructor Create (AOwner : TComponent);override;

procedure Click; override;
В раздел реализации включите программные коды новых методов:
constructor TNew1Button.Create(AOwner : TComponent);

      begin

         inherited  Create(AOwner); {вызывается конструктор базового класса}

          FCount:=0;                           {счетчик нажатий обнуляется}
      end;

    procedure TNew1Button.Click;

        begin

           inherited Click;                  {вызывается стандартный обработчик события OnClick}

           Inc(FCount);                        {выполняются действия нового обработчика}

           Caption:=IntToStr(FCount);

         end;
Установите компонент New1Button на вкладку New палитры компонентов Delphi, выполнив действия пункта 4 примера 1. 

Выполните тестирование нового компонента, для чего создайте новое приложение, поместите на форму компонент New1Button и другие стандартные компоненты (кнопки, метки и т. д.) для обеспечения приемлемого интерфейса приложения. Проверьте, что новая кнопка действительно считает нажатия на себя.
Выполнив действия пункта 6 (пример 1), подготовьте и зарегистрируйте в среде Delphi значок для компонента New1Button. 
Модифицируйте тестовое приложение, создав в нем собственный обработчик события OnClick для кнопки New1Button (например, в обработчике можно вывести какое-либо сообщение процедурой ShowMessage или иным способом). Проверьте работу приложения и сделайте соответствующие выводы.
1.6. Создание и добавление свойств в компонент
Наиболее сложным и ответственным этапом в создании нового компонента является создание и добавление в него новых свойств. 
Концепция свойств является одной из самых важных особенностей объектной модели языка Object Pascal. Без понимания тонкостей работы со свойствами трудно создать более или менее сложный пользовательский компонент в Delphi.
1.6.1. Понятие свойства. Классификация свойств

Свойства аналогичны обычным переменным экземпляра класса и служат для доступа к внутренним полям класса. Свойство можно трактовать как средство для чтения и записи данных в объекте. Функция чтения свойства обязана возвращать какой-либо результат, а процедура или функция записи свойства меняет состояние объекта. Свойство обязано иметь идентификатор и тип, и при обращении из рабочей копии класса свойство выглядит как переменная. Но в отличие от переменных, которые являются простым хранилищем некоего значения, которое может быть прочитано и изменено, свойства позволяют выполнять дополнительные действия над значениями полей, влияющие на внутреннее состояние объекта. Это позволяет использовать свойства для контроля изменения значений в полях класса. Кроме того, в отличие от обычной переменной свойство не может быть использовано в качестве фактического параметра метода, если его формальный параметр является параметром-переменной, т. е. имеет описатель var.

Описание свойства имеет следующий обобщенный синтаксис:
property <имя свойства> : <тип> [index <целая константа>] <спецификаторы свойства>
При объявлении свойства используется зарезервированное слово property, после которого указываются:

1. Имя свойства, которое используется для ссылок на значение свойства. В результате свойства получают внешний вид полей данных.
2. Тип свойства, для этого используется символ двоеточие после имени свойства.
3. Директива index (необязательная), позволяющая осуществлять раздельный доступ двух и более свойств к одному и тому же полю объекта. Доступ организуется в соответствии со значением целой константы.
4. Спецификаторы (описатели) свойства – это дополнительные директивы read, write, stored, default (или nodefault).
Директива read используется для указания метода, который будет использоваться для осуществления выборки значения свойства. Метод должен быть функцией, чей возвращаемый тип является тем же самым, что и тип свойства.
Вместо метода доступа для чтения можно просто указать внутреннее поле хранения данных. Подобная форма записи приводит к тому, что значение свойства извлекается прямо из внутреннего поля данных.
Директива write определяет, какой метод будет использоваться для присвоения свойству значения. Метод должен быть процедурой, имеющей единственный параметр, тип кото​рого должен совпадать с типом свойства.
При обращении к значению свойства происходит перена​правление на соответствующий метод. Например, оператор s:=Editl.Text; автоматически будет преобразован в оператор s:=Editl.GetText; а оператор Editl.Text: = 'Test' — в опе​ратор Editl.Text('Test' ).
Описание свойства должно содержать директиву read или write или обе сразу. Если описание свойства включает в себя только директиву read, то оно является свойством только для чтения. В свою очередь, свойство, чье описание включает в себя только директиву write, является свойством только для записи. При присвоении свойству, определенному с директивой только для чтения, какого-либо значения или при использова​нии в выражении свойства с директивой только для записи всегда возникает ошибка.
Когда программист использует Инспектор объектов для изменения свойств формы или свойств компонентов, то результирующие изменения заносятся в файл формы. Файлы форм представляют собой файлы ресурсов Windows, и когда приложение запускается, то описание формы подгружается из этого файла. Для определения того, что должно сохраняться в файле формы, служат специфика​торы памяти — необязательные директивы stored, default и nodefault. Эти директивы влияют на информацию о типе во время вы​полнения, генерируемую для свойств published.
Директива stored управляет тем, будет или нет свойство действительно запоминаться в файле формы. За директивой stored должны следовать либо константы True или False, либо имя поля, имеющего тип Boolean, либо имя метода, у которого нет параметров, и возвращающего значение типа Boolean. Например,
property Hint: string read FHint write FHint stored IsHintStored;
Если после директивы stored установлено значение False или это значение возвращается соответствующим методом, то значение свойства не запоминается в файле формы, и сохраняется в противном случае. 
Если свойство не содержит директиву stored, то оно рассматривается как содержащее ее с параметром True.
Директивы default и nodefault управляют значениями свойства по умолчанию. За директивой default должна следовать константа того же типа, что и свойство, например:
property Tag: Longint read FTag write FTag default 0;
Чтобы перекрыть наследуемое значение default без указания нового значения, используется директива nodefault. Директивы default и nodefault работают только с порядковыми типами и множествами, нижняя и верхняя границы которых лежат в промежутке от 0 до 31. Если такое свойство описано без дирек​тив default и nodefault, то оно рассматривается как с директи​вой nodefault. Для вещественных типов, указателей и строк значение после директивы default может быть только 0, NIL и ‘’(пустая строка) соответственно.
Когда Delphi сохраняет компонент, то просматриваются спецификаторы памяти published свойств компонента. Если значение текущего свойства отличается от default значения (или директива default отсутствует) и параметр stored равен True, то значение свойства сохраняется в файле формы, иначе свойство не сохраняется.
Свойства в объектной модели Delphi можно классифицировать по четырем различным признакам:
· по методу доступа к полю, который обеспечивает свойство;
· по использованию различных индексов;
· по способу сохранения значений в файле формы;
· по типу.
Сделанная по указанным признакам полная классификация свойств представлена рис.12.

Рис. 12
Остановимся на рассмотрении нескольких свойств, наиболее часто встречающихся у компонентов Delphi. Описания остальных можно найти в [1, 2, 3].
1.6.2. Простые свойства
Свойства этого типа наиболее часто встречаются у компонентов Delphi. Простое свойство может быть числовым, строковым или символьным. Допускается непосредственное редактирование значений этих свойств. Примером простых свойств являются: Caption, Text, Tag, Left, Top.
В соответствии с названием создание простых свойств несложно. 
Рассмотрим следующий пример:

type
TMyComponent = class(TCustomControl) 
private
{ Private declarations }
fInt:Integer;
fStr:String;
fCh:Char; 
protected
{ Protected declarations } 
public
{ Public declarations } 
published
{ Published declarations }
property Int:Integer read fInt write fInt;
property Str:String read fStr write fStr;
property Ch:Char read fCh write fCh; 
end;
В секции частных объявлений указаны три поля объекта: целочисленное, строковое и символьное
. Для обеспечения доступа к этим полям в секции published объявлены три соответствующие свойства. Если такой компонент окажется на форме, то внешний вид Инспектора объектов будет соответствовать рис. 13. 

Рис. 13
Следует отметить, что помимо новых свойств Int, Str, Ch в Инспекторе объектов отображены свойства, унаследованные от родительского класса 
1.6.3. Перечислимые свойства
Свойства перечислимого типа разрешают выбор значения только из списка, предопределенного разработчиком компонента. Кроме того, в качестве перечислимых могут выступать свойства логического типа. Примером таких свойств являются: Color, DragKind, DragMode, FormStyle. Признаком перечислимого свойства в Инспекторе объектов является появление ниспадающего списка с допустимыми значениями справа от имени свойства.
Перечислимое свойство может создаваться как на ос​нове уже существующих перечислимых типов (TColor, TCursor и т. п.), так и на базе нового типа, описанного в процессе разработки компонентов. 
В следующем примере описывается класс TWeekDay, включающий в себя семь значений дней недели: 
type TWeekDay =(wdMon, wdTu, wdWe, wdTh, wdFr, wdSa, wdSu); 
type
TMyComponent = class(TCustomControl)
private
{ Private declarations }
fDay:TWeekDay; 
published
{ Published declarations }
property Day:TWeekDay read fDay write fDay; 
end;
По принятой договоренности имена элементов перечислимого типа должны начи​наться с двух строчных символов, указывающих на принадлежность к данному классу. В данном случае это символы «w» и «d». На этом примере можно отметить две особенности свойств данного типа:

1) в тексте класс TWeekDay описывается перед объявлением класса разрабатываемого компонента;
2) в ниспадающем списке Инспектора объектов возможные значения перечислимого свойства отображаются не в той последовательности, в которой они были описаны в тексте, а в алфавитном порядке (рис. 14).

Рис. 14
1.6.4. Свойство типа множество
В Инспекторе объектов рядом с именем свойства типа множество отображается символ «+». В качестве примера такого свойства может выступить свойство формы BorderIcons. Двойной щелчок по свойству разворачивает в Инспекторе объектов дополнительные строки, изменяя значения которых пользователь определяет состав множества. Множество допускается редактировать и вручную путем изменения его состава внутри квадратных скобок в строке ввода.
Преимущество использования свойства типа множество над другими типами свойств заключается в предоставляемой возможности одновременного выбора нескольких значений из состава множества. Для пояснения свойства такого типа используем исходный код предыдущего примера с днями недели. 
Дополним текст строкой с объявлением типа данных TSetWeekDay, являющимся множеством от TWeekDay, затем объявим поле fSetWeekDay и соответствующее ему свойство SetWeekDay:
type TWeekDay =(wdMon, wdTu, wdWe, wdTh, wdFr, wdSa, wdSu); 
type TSetWeekDay = set of TWeekDay;
type
TMyComponent = class(TCustomControl) 
private
{ Private declarations }
fSetWeekDay : TSetWeekDay; 
protected
{ Protected declarations } 
public
{ Public declarations } 
published
{ Published declarations }
property SetWeekDay:TSetWeekDay read fSetWeekDay write fSetWeekDay; 
end;
На рис. 15 представлен вид Инспектора объектов, отображающего компонент MyComponenti со свойством SetWeekDay.
[image: image15.png]:Lbiw Incpecto





Рис. 15

1.6.5. Свойство типа массив
Свойство типа массив может принимать любой тип данных, определенный в Инспекторе объектов, и обозначается в Инспекторе объектов кнопкой с многоточием (например, свойства Items, Lines). В них списки строк хранятся в массивах, доступ к которым обеспечивается известными методами языка Object Pascal.
Первым шагом для создания свойства-массива в новом компоненте должно стать объявление массива с указанием его имени, количества элементов и типа хранимых данных, например:
type TMyIntArray = Array [0..9] of integer;
Далее в секции private описывается внутреннее поле fMyIntArray компонента типа TMyIntArray. Но затем обычная последовательность описания свойств будущего компонента нарушается. Это связано с тем что, Инспектор объектов не приспособлен к отображению свойств типа массив. Поэтому попытка разместить такое свойство в секции published не принесет никаких результатов.
Ответ на вопрос о том, каким образом в Инспекторе объектов оказывается свойство-массив (например, Lines у компонента ТМеmо), можно найти в справочной системе Delphi, в разделах, посвященных созданию редактора свойств и редактора компонента. Обычно доступ к полю реализуется в секции public:
type
TMyComponent = class(TCustomControl)
private
fMyIntArray:TMyIntArray;
protected
function GetMyArrayUnit(ind : Integer) :Integer;
procedure SetMyArrayUnit(ind, value : Integer); 
public
property MyArrayUnit[ind:Integer]:Integer read GetMyArrayUnit write 
SetMyArrayUnit;
published
end;
… 
implementation
…
function TMyComponent.GetMyArrayUnit(ind:Integer):Integer;
begin
// Чтение элемента с индексом ind из массива
Result:=fMyIntArray[ind];
end;
procedure TMyComponent.SetMyArrayUnit(ind,value:Integer);
begin
// Запись значения value в элемент массива с индексом ind
fMyIntArray[ind]:=value;
end;
В этом примере чтение из массива обеспечивается функцией GetMyArrayUnit(ind:Integer): Integer, а запись в массив - процедурой SetMyArrayUnit(ind, value:Integer)
. В качестве параметра в обоих методах обязательно присутствует индекс элемента в массиве - ind.
Таким образом, при создании свойства типа массив оно объявляется в секции public. Для организации доступа к элементам массива необходимо описание как минимум двух внутренних методов компонента, отвечающих соответственно за чтение и запись в массив.
Свойствами-массивами не поддерживаются спецификаторы памяти, а директива default при описании свойства-массива имеет другое назначение [2].

Если свойство создается на основе многомерного массива, то последовательность индексов в методах доступа должна точно соответствовать очередности объявления индексов у свойства.
1.6.6. Добавление свойств в компонент

Для того чтобы дополнить создаваемый компонент новым свойством, надо выполнить следующие шаги:
1. Определить название и тип свойства.
2. Объявить внутреннее поле объекта выбранного типа.
3. Объявить свойство.
4. Определить методы чтения/записи свойства (если они используются для доступа к значениям свойства).
5. При необходимости установить значение по умолчанию.
6. Определить средства сохранения/загрузки значения свойства.
Рассмотрим содержание этих шагов.

Для добавления поля к объекту в секции private класса объявляется переменная необходимого типа. Предположим, что в состав VCL входит некий гипотетический класс, описывающий автомобиль, - TAutomobile. Ставится задача создать компонент-наследник TNewAutomobile, имеющий собственное свойство Speed (скорость) целого типа. Объявление поля объекта с именем fSpeed следует поместить в секцию частных объявлений:
type
TNewAutomobile = class(TAutomobile) 
private
{ Private declarations } 
fSpeed:cardinal;         //объявление поля объекта целого типа
end;
Для того чтобы свойство отобразилось в Инспекторе объектов, необходимо объявить его в секции published и указать способ выполнения операций чтения/записи данных в поле объекта:
type
TNewAutomobile = class(TAutomobile) 
private 
fSpeed: cardinal;
published
property Speed: cardinal read fSpeed write fSpeed; 
end;
В составе строки объявления свойства присутствуют два ключевых слова read (чтение) и write (запись). Они указывают на способ чтения и записи значения свойства. В данном примере операции чтения и записи перенаправляются сразу в поле. Если отказаться от одного из ключевых слов, то будет создано свойство, доступное только для чтения или только для записи.
Простейший вид взаимодействия свойства и соответствующего ему поля называется прямым доступом. Приведенный пример демонстрирует использование прямого доступа при организации операций чтения/записи. Здесь любое изменение значения свойства Speed сразу передается в поле fSpeed. Но на практике прямой доступ применяется не часто.
При проектировании нового компонента обычно используется более универсальный способ доступа к значениям свойства, основанный на использовании методов чтения и записи значений свойств. Эти методы объявляются в секции private после описания поля объекта:
type
TNewAutomobile = class(TCustomControl)
private
fSpeed: cardinal;
procedure SetSpeed(Value:cardinal);  //установка значения в поле 
function GetSpeed:cardinal;                //чтение значения из поля
published
property Speed: cardinal read GetSpeed write SetSpeed; 
end;
Метод, отвечающий за запись, всегда является процедурой и, как правило, обладает единственным параметром. Второй параметр понадобится только при работе со свойством типа массив. Название параметра (Value) может быть любым, но обычно используют Value
. Тип параметра должен совпадать с типом свойства. В качестве параметра Value внутреннему полю объекта передается новое значение. Например, в процедуре может проверяться корректность вводимых данных и в случае успеха полю присваивается новое значение:
procedure TNewAutomobile.SetSpeed(Value:cardinal);
begin
if Value<=300 then fSpeed:=Value else ShowMessage(‘Недопустимое значение');
end;
Метод, отвечающий за чтение, всегда оформляется в виде функции, которой не передаются параметры. Единственным исключением будут функции, осуществляющие чтение из свойства типа массив (в этом случае потребуется параметр, определяющий индекс элемента в массиве) и чтение с использованием команды index. Результат, возвращаемый функцией, должен иметь такой же тип, как и у соответствующего свойства:
function TNewAutomobile.GetSpeed:cardinal;
begin
Result:=fSpeed;
end;
Методы для чтения/записи свойства могут быть статическими и виртуальными, но не динамическими. 
Свойство может быть смешанным. Это означает, что при чтении свойства вызывается метод, а при записи обращаются к переменной и наоборот:

property Data:integer read GetData write FData;

property Data:integer read FData write SetData;

Одной из возникающих при разработке новых компонентов задач является публикация свойства, скрытого в родительском классе. Например, потомки класса TWinControl наследуют более 70 свойств своего предка, но по умолчанию в Инспекторе объектов отображается всего лишь 9: Cursor, Height, HelpContext, Hint, Left, Name, Tag, Top и Width. Для того чтобы сделать унаследованные свойства доступными во время разработки, необходима повторная декларация этих свойств, но уже в секции published исходного кода компонента-потомка:
type
TMyComponent = class(TWinControl)
…
published
property Align;        //свойство унаследовано от TControl
end;
Примечание.
Как уже упоминалось ранее, Инспектор объектов обеспечивает доступ только к свойствам, объявленным в секции published. Свойства, опубликованные в секции public, будут доступны только из кода приложения. Кроме того, при работе со свойствами следует учитывать две особенности компилятора Delphi, а именно: 

1) при объявлении нового свойства с именем, совпадающем с именем уже имеющегося свойства, ранее определенное свойство «затеняется»; 

2) свойства, которые имеют доступ только для чтения или только для записи, не отображаются в инспекторе объектов, даже если они объявлены в секции published. 

Программные коды методов чтения/записи свойства размещаются в разделе реализации модуля компонента как обычные процедуры.

Установка значений свойств по умолчанию и задание средств сохранения/загрузки выполняется с помощью директив default, nodefault и stored (см. п. 1.6.1).
Пример 4. Создать компонент, который представляет собой модифицированную кнопку быстрого доступа SpeedButton. Модификация состоит в том, что в дополнение к стандартному изображению кнопки должны рисоваться внутренняя и внешняя рамки выбираемых цветов.
Решение данной задачи сводится к тому, что для обеспечения возможности настройки интерфейса кнопки следует:

1) создать два свойства: одно для представления цвета внутренней рамки, а другое – для цвета внешней рамки;

2) переопределить метод рисования компонента, в котором в дополнение к изображению кнопки SpeedButton будут рисоваться две рамки: внешняя и внутренняя.
Процесс разработки модуля компонента, как и во всех предыдущих примерах, начинается с вызова мастера компонента. В окне мастера выберите базовый класс TSpeedButton, задайте имя нового компонентного класса, например, TFramedButton. Остальные параметры можно оставить такими же, как в предыдущих примерах. Выполнив действия пункта 3 в примере 1, создайте заготовку модуля нового компонента.
Разработку программы модуля компонента следует начать с определения атрибутов новых свойств компонента и их описания.
Пусть цвет внутренней рамки определяется свойством с именем InFrameColor, а цвет внешней рамки – свойством OutFrameColor. Оба свойства принадлежат классу TColor, т. е. их значениями являются различные цвета. Свойства будут создаваться с возможностью чтения и записи их значений в специально отводимых полях FInFrameColor и FOutFrameColor. Будем считать, что чтение выполняется прямым обращением к полю, а запись – с помощью методов SetInFrameColor и SetOutFrameColor. Описания полей и методов размещаются в разделе private класса TFramedButton, и будут иметь следующий вид:
type

  TFramedButton = class(TSpeedButton)

  private

  FInFrameColor:TColor;

  FOutFrameColor:TColor;

  procedure SetInFrameColor(Value:TColor);

  procedure SetOutFrameColor(Value:TColor);

Программные тексты процедур SetInFrameColor и SetOutFrameColor помещаются в раздел implementation модуля и имеют незначительные отличия. Процедура SetInFrameColor представляется следующим образом:

procedure TFramedButton.SetInFrameColor(Value:TColor);

begin

If FInFrameColor<>Value then        {если значение действительно новое}

Begin

FInFrameColor:=Value;               {записывается новое значение свойства}

Invalidate;                       {так как свойство изменилось компонент перерисовывается}

end;

end;
Процедура SetOutFrameColor имеет аналогичный текст. 
Для воспроизведения на новом компоненте внешней и внутренней рамок, как было отмечено выше, необходимо переопределить метод рисования компонента. Таким методом является метод Paint. Переопределяя этот метод для уже реализованных графических компонентов необходимо вызвать унаследованный родительский метод Paint. Для рисования на кнопке двух рамок можно использовать средства графики классов TCanvas и TBrush. Переопределение метода рисования для создаваемого компонента будет выглядеть следующим образом:
procedure TFramedButton.Paint;

begin

inherited Paint;                                  {вызов родительского метода}

Canvas.Brush.Color:=OutFrameColor;

Canvas.FrameRect(rect(4,4,width-4,height-4));   {рисование внешней рамки}
Canvas.Brush.Color:=InFrameColor;

Canvas.FrameRect(rect(10,10,width-10,height-10));    {рисование внутренней рамки}
end;
Здесь рамки представляются в виде прямоугольников, для рисования которых используется метод FrameRect. Область прямоугольника задается с помощью процедуры rect, в которую передаются координаты x, y верхнего левого и правого нижнего углов прямоугольника.
Метод Paint следует описать как переопределяемый (т. е. с директивой override) в разделе protected создаваемого модуля, а текст поместить в раздел implementation. 
Установите компонент FramedButton на вкладку New палитры компонентов Delphi, выполнив действия пункта 4 примера 1. 

Выполните тестирование нового компонента, для чего создайте новое приложение, поместите на форму созданный компонент и другие стандартные компоненты (кнопки, метки и т. д.) для обеспечения приемлемого интерфейса приложения. Проверьте работу приложения, задавая различные значения цветов внешней и внутренней рамок кнопки с помощью Инспектора объектов.
Подготовьте и зарегистрируйте в среде Delphi значок для компонента FramedButton. 
1.7. Создание справки компонента
Процедура создания справки компонента включает следующие этапы. Первым этапом является подготовка файла справки *.hlp и файла содержания справки *.cnt с помощью стандартного для Windows компилятора файлов помощи hcw.exe (см. лабораторную работу № 5).
На втором этапе файл помощи и файл содержания справки устанавливаются в систему подсказок Delphi.

Установка выполняется следующим образом. В меню системы выполняется команда Help/Customise. На экране откроется диалоговое окно настройки системы подсказки Delphi (рис. 16).
На вкладке Contents с помощью кнопки [image: image16.png]


 (Add Files) следует добавить файл содержания справки *.cnt. На вкладках Index и Link с помощью той же кнопки необходимо добавить файл помощи *.hlp. В заключение следует сохранить настройки системы подсказки, выполнив в окне 
[image: image17.png]File Edit Help

=lolx|

=R

{Corferis | index | Link | Project]

Rave.cnt

Title [Path Size [ Dete

@ What's Newin Delphi 7 CiProgram Files\Borland), KB 08/03/02 0.0
@ Delphi Help CiProgram Files\Borland, KB 08/03/02 030
@ Programming with Delphi CiPragram Files\Borland,..  35KB 08/09/02 0900
@ Developing Database Appli.. C\Program Files\Borlanch..  33KE 08/03/02 0300
@ Creating Internet and Intranet .. CAProgram Files\Borlanch..  14KE 08/03/02 0300
@ Creating Custom Camponents  CAProgram Files\Borlanch..  16KE 08/03/02 0300
@ Developing COM-based app... CAProgram Files\Borlanch..  10KE 08/03/02 0300
@ Delphi Object and Compone... CAProgram Files\Borlanc.  368KE 08/03/02 0300
@ InterBase Express Reference  C\Program FilesiBorlandh, KB 08/03/02 030
@ Delphi Language Reference  CAProgram Files\Borlanc.  18KE 08/03/02 0300
@ Delphi Procuctivity Tools  CProgram FilesiBorlandh, KB 08/09/02 0300
@ incly CiPragram Files\Borland,..  308KB  08/09/02 0900
@ Rave Developers Guide CiPragram Files\Borland,.  22KB  08/09/02 0900





Рис. 16

OpenHelp команду File/Save Project As….После выполнения этих действий созданный файл помощи становится частью системы подсказки Delphi.
1.8. Удаление компонентов из среды разработки
Для удаления компонента командой меню File/Open… или File/Open Project… вначале следует открыть окно с пакетом MyPackage.dpk, содержащим данный компонент (рис. 17). 

Рис. 17
В окне пакета нажать кнопку [image: image19.png]Remove



 и в открывшемся диалоговом окне Remove From Project (рис. 18) щелчком мыши выделить исключаемые из пакета файлы и нажать ОК.

Рис.18
Затем подтвердить удаление в открывшемся окне запроса. В заключение необходимо перекомпилировать пакет с тем, чтобы сделанные изменения были зафиксированы в среде разработки. Для этого в окне пакета нажимается кнопка [image: image21.png]Compile



. После перекомпиляции появится информационное окно (рис. 19), сообщающее об удаленных из среды разработки компонентах.

Рис. 19
После этого вкладка с соответствующим компонентом удаляется из палитры компонентов Delphi.
2. Практическая часть

2.1. Выполните все примеры, рассмотренные в первой части. 
2.2. Модифицируйте пример 4 так, чтобы:
a) обеспечивалась возможность задавать координаты x, y верхнего левого угла внешней и внутренней рамки с помощью свойств;
b) обе рамки рисовались линиями задаваемым цветом и толщиной.
2.3. Задания для самостоятельного выполнения 
a) разработайте компонент NewEdit, который позволяет выравнивать текст по правой границе текстового поля;
b) создайте компонент RunText, реализующий «бегущую» текстовую строку с регулируемой скоростью перемещения текста;
Примечание. 1. За основу взять компоненты Label и Timer.
2. Для компонента RunText разработать и установить справку.
c) разработайте компонент LichtButton, реализующий кнопку, при наведении курсора мыши на которую, а также нажатиях/отпусканиях кнопок мыши будет меняться ее цвет. Цвет рамки кнопки и ее ширина должны задаваться.
Примечание. При проектировании компонента создать свойства, отвечающие за:

1) цвет кнопки, когда происходит событие мыши OnMouseDown;
2) начальный цвет кнопки (при событии мыши OnMouseUp);

3) цвет кнопки, когда происходит событие мыши OnMouseMove;
4) цвет рамки кнопки и ее ширину.
3. Содержание отчета по лабораторной работе
В отчете по лабораторной работе по пп.2.1 – 2.3 должны быть приведены распечатки текстов и форм в моменты выполнения всех приложений, в которых проверяются разработанные компоненты.

Примечание.
После защиты лабораторной работы и получения зачета по ней по указанию преподавателя следует удалить из среды Delphi все новые компоненты.

� Библиотека CLX (Component Library for Cross Platform) является дальнейшим развитием библиотеки VCL для разработки кроссплатформенных приложений, которые могут работать как под ОС Windows, так и Linux.


� Файлы *.dcp – это аналог скомпилированных модулей типа dcu для обычной программы.


� можно использовать и другие варианты реализации данных методов


� По существующим среди программистов соглашениям в Delphi имя поля всегда должно начинаться с префикса «f» (от англ. field – поле).


� Согласно существующим соглашениям имена процедур, отвечающих за запись значений в поле, должны начинаться с приставки Set, а имена функций, обеспечивающих чтение, - с приставки Get. Вторая часть имени включает название свойства, с которым работают эти методы, например SetMyArrayUnit(), GetMyArrayUnit().





� value (англ.) - величина





